Base-line Correction of Accelerogram—CRAC

The program CRAC (Base-line **Co**r**r**ection of **Ac**celerogram) is a subroutine subprogram that corrects for a given acceleration time history by adjusting the baseline so that the velocity at the end of the duration is zero and the residual displacement is not unreasonably excessive.

CRAC（Base-line **Co**r**r**ection of **Ac**celerogram）

【Purpose】

To correct the baseline and modify the acceleration time history so that the velocity is zero at the end of the duration and the residual displacement is within a reasonable value.

【Usage】

(1)　How to connect

```
CALL CRAC (DT, NN, DDYMAX, DDY, ND, UW1, UW2)
```

Argument	Type	Parameter in calling program	Return Parameter
DT	R	Time interval (unit : sec)	Unchanged
NN	I	Total number of real data DDY,DY,Y	Unchanged
DDYMAX	R	Maximum value of input acceleration (unit : Gal)	Unchanged
DDY	R 1-D array (ND)	Original acceleration time history (unit : Gal)	Corrected acceleration time history (unit : Gal)
ND	I	Dimension size of DDY, UW1, UW2 in calling program	Unchanged
UW1	R 1-D array (ND)	No need to input here	(working area)
UW2	R 1-D array (ND)	No need to input here	(working area)

(2)　Necessary subroutines and function subprograms

IACC

【Calculation Method】

First, the given acceleration time history $\ddot{y}(t)$is integrated by using the subroutine **IACC** and obtains the velocity and displacement time histories $\dot{y}(t)$ and $y(t)$. Then, the modified values of displacement, velocity, and acceleration $\hat{y}(t), \hat{\dot{y}}(t), \hat{\ddot{y}}(t)$ shall be expressed as follows.

$$\left.\begin{aligned} \hat{y}(t) &= y(t) - \left(\frac{1}{2}a_0 t^2 + \frac{1}{6}a_1 t^3\right) \\ \hat{\dot{y}}(t) &= \dot{y}(t) - \left(a_0 t + \frac{1}{2}a_1 t^2\right) \\ \hat{\ddot{y}}(t) &= \ddot{y}(t) - \left(a_0 + a_1 t\right) \end{aligned}\right\} \qquad \text{(a)}$$

If the duration is T, the condition that $\hat{\dot{y}}(t) = 0$ in the second equation of Eq. (a) is as follows.

$$a_0 = \frac{\dot{y}(T)}{T} - \frac{a_1 T}{2} \qquad \text{(b)}$$

Thus, we get

$$\frac{da_0}{da_1} = -\frac{T}{2} \qquad \text{(c)}$$

In order to satisfy the condition that the residual displacement $\hat{y}(T)$ at t=T is not unreasonably excessive, the coefficients of the cubic polynomial in parentheses on the right-hand side of the first equation of Eq. (a) are obtained to best fit the curve $y(t)$ by using the following least-squares method.

$$\varepsilon = \int_0^T \left[y(t) - \left(\frac{1}{2}a_0 t^2 + \frac{1}{6}a_1 t^3\right)\right]^2 dt$$

and

$$\frac{d\varepsilon}{da_1} = 0 \qquad \text{(d)}$$

From Eqs. (b), (c), and (d), the coefficient $a_1$can be obtained as follows.

$$a_1 = \frac{28}{13}\cdot\frac{1}{T^2}\left[2\dot{y}(T) - \frac{15}{T^5}\int_0^T y(T)(3Tt^2 - 2t^3)dt\right] \qquad \text{(e)}$$

In this program, the integration of the right-hand side of Eq. (e) is performed using the simplest trapezoidal rule, since it does not require particularly high accuracy.

If a_1 is determined by Eq. (e), then a_0 is also determined by Eq. (b), and the corrected acceleration time history $\hat{\ddot{y}}(t)$ is calculated by the third equation in Eq. (a).

However, since the maximum value of the corrected acceleration time history $\hat{\ddot{y}}(t)$ is slightly different from the maximum value of the original time history $\ddot{y}(t)$, the entire corrected time history $\hat{\ddot{y}}(t)$ is multiplied by a factor C to restore the maximum acceleration to its original value. Where C is the following.

$$C = |\ddot{y}(t)|_{\max} / |\hat{\ddot{y}}(t)|_{\max}$$

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *
C     SUBROUTINE FOR BASE-LINE CORRECTION OF ACCELEROGRAM
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *
C
C                                       CODED BY Y.OHSAKI
C
C     PURPOSE
C       TO CORRECT THE ORIGINAL ACCELERATION TIME HISTORY BY BASE-LINE
C       ADJUSTMENT SO THAT (1) THE TERMINAL VELOCITY VANISHES, AND (2)
C       THE PERMANENT DISPLACEMENT CONVERGES WITHIN A REASONABLE LIMIT
C
C     USAGE
C       CALL CRAC(DT,NN,DDYMAX,DDY,ND,UW1,UW2)
C
C     DESCRIPTION OF ARGUMENTS
C       DT      - TIME INCREMENT IN SEC
C       NN      - TOTAL NUMBER OF DATA IN ACCELERATION TIME HISTORY
C       DDYMAX  - MAX. ACCELERATION IN GALS
C       DDY(ND) - ORIGINAL/CORRECTED ACCELERATION TIME HISTORY IN GALS
C                 AT CALL/RETURN
C       ND      - DIMENSION OF DDY,UW1,UW2 IN CALLING PROGRAM
C       UW1(ND) - WORKING AREA
C       UW2(ND) - WORKING AREA
C
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C       IACC
C
      SUBROUTINE CRAC(DT,NN,DDYMAX,DDY,ND,UW1,UW2)
C
      DIMENSION DDY(ND),UW1(ND),UW2(ND)
C
      CALL IACC(DT,NN,DDY,UW1,UW2,ND,DUMMY,DUMMY)
      TT=REAL(NN-1)*DT
      T=0.
      DO 110 M=1,NN
      UW2(M)=UW2(M)*(3.*TT-2.*T)*T**2
      T=T+DT
  110 CONTINUE
      SUM=(UW2(1)+UW2(NN))/2.
      DO 120 M=2,NN-1
      SUM=SUM+UW2(M)
  120 CONTINUE
      SUM=SUM*DT
      A1=28./13./TT**2*(2.*UW1(NN)-15./TT**5*SUM)
      A0=UW1(NN)/TT-A1/2.*TT
      T=0.
      ACMAX=0.
      DO 130 M=1,NN
      DDY(M)=DDY(M)-A0-A1*T
      ACMAX=AMAX1(ACMAX,ABS(DDY(M)))
      T=T+DT
  130 CONTINUE
```

```
      COEF=DDYMAX/ACMAX
      DO 140 M=1,NN
      DDY(M)=DDY(M)*COEF
  140 CONTINUE
      RETURN
      END
```